> restart; 

> with(plots):
with(linalg):
with(plottools): 

Displaymasters: Liquid Crystal Displays, Theory and Modelling. 

this maple file available at http://www.maths.strath.ac.uk/~aas99102/displaymasters 

example energy surface 


the system (represented by the red dot) will prefer to sit in an energy minimum. 

> 
plotsetup(default);
display(plot3d(1-cos(n)^2*(cos(v))^2,n=-2..2,v=-2..2,axes=framed,view=[DEFAULT,DEFAULT,-0.2..1]),sphere([0,0,0.11],0.1,color=red, style=patchnogrid)); 

PRIVATE "TYPE=PICT;ALT=[Maple Plot]"

dynamically, the system will behave as a ball rolling on a surface. 

> display(seq(display(plot3d(1-cos(n)^2*(cos(v))^2,n=-2..2,v=-2..2,axes=framed,view=[DEFAULT,DEFAULT,-0.2..1.2]),sphere([0,-(Pi/2)*exp(-0.1*i)*cos(0.5*i),1-cos(-(Pi/2)*exp(-0.1*i)*cos(0.5*i))^2+0.11],0.1,color=red, style=patchnogrid)),i=0..30),scaling=constrained,insequence=true); 

PRIVATE "TYPE=PICT;ALT=[Maple Plot]"

> 
nematic elastic energy 


Example: the director lies in the xz-plane 
> n:=vector([cos(theta(z)),0,sin(theta(z))]);
xvec:=vector([x,y,z]); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> dn:=diverge(n,xvec); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> cn:=curl(n,xvec); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


the K_1, splay term 

> splay:=(K_1/2)*dn^2; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


the K_2, twist term 

> twist:=(K_2/2)*dotprod(n,cn)^2; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


the K_3, bend term 

> bend:=(K_3/2)*multiply(crossprod(n,cn),crossprod(n,cn)); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

the K_2+K_4, saddle-splay term 

> saddlesplay:=((K_2+K_4)/2)*diverge(vector([seq(sum('n[i]*diff(n[j],xvec[i])','i'=1..3)-(diverge(n,xvec))*n[j],j=1..3)]),xvec); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


total energy density 

> w[d]:=splay+twist+bend+saddlesplay; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


one constant approximation 

> w_oneconst[d]:=simplify(subs({K_1=K,K_2=K,K_3=K,K_4=0},w[d])); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> 
nematic magnetic energy density 

> H:=vector([0,0,H_3]); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


magnetic energy density 

> w[m]:=-chi[perp]*H^2/2-chi[a]*multiply(H,n)^2/2; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> 
minimisation 


total energy density 
> w[total]:=w_oneconst[d]+w[m]; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> w_simp:=subs(theta(z)=th,subs(diff(theta(z),z)=dth,w[total])); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

differentiate w by the gradient term 

> temp:=subs({th=theta(z),dth=diff(theta(z),z)},diff(w_simp,dth)); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


then differentiate by z 
> term1:=diff(temp,z); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


differentiate w by the theta 

> term2:=subs({th=theta(z),dth=diff(theta(z),z)},diff(w_simp,th)); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


Euler-Lagrange equation 
> ELequation:=term1-term2; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> 
> 
FLC model 


elastic torque 

> Q[k]:=(B/d^2)*(sin(2*phi)-2*cos(phi_chev)*sin(phi)); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


dielectric torque 

> Q[e]:=(epsilon_0*(deltaeps-Deps*sin(theta)^2))*V^2/(2*d^2)*cos(delta)^2*sin(2*phi)+(epsilon_0*Deps*sin(2*theta)*V^2)/(2*d^2)*sin(delta)*cos(delta)*sin(phi); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


spontaneous polarisation torque 

> Q[p]:=P[s]*V/d*cos(delta)*cos(phi); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


model parameters 

> pars:={epsilon_0=8.845e-12,Deps=-5,deltaeps=1,B=1.0e-11,theta=evalf(25*Pi/180),delta=evalf(22*Pi/180),P[s]=5e-5,d=1e-6,mu=1e-4}:
phichev:=evalf(subs(pars,arccos(tan(delta)/tan(theta)))):
pars2:=pars union {phi_chev=phichev}; 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

> torque:=evalf(subs(pars2,Q[k]+Q[e]+Q[p]));
ktorque:=evalf(subs(pars2,Q[k])):
etorque:=evalf(subs(pars2,Q[e])):
ptorque:=evalf(subs(pars2,Q[p])): 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

PRIVATE "TYPE=PICT;ALT=[Maple Math]"


plot the corresponding energies for the elastic, dielectric and polarisation effects at a fixed voltage V 
> Vsub:=-0.1:

#plot({int(-ktorque,phi)},phi=-0.8..0.8,color=black,title="chevron_energy");
display(plot({int(subs(V=Vsub,-ktorque),phi)},phi=-Pi..Pi,color=red),plot({int(subs(V=Vsub,-ptorque),phi)},phi=-Pi..Pi,color=blue),plot({int(subs(V=Vsub,-etorque),phi)},phi=-Pi..Pi,color=green),title="blue=spont.pol.,red=elastic,green=dielectric");
plot({int(subs(V=Vsub,-torque),phi)},phi=-Pi..Pi,color=black,title="total_energy"); 

PRIVATE "TYPE=PICT;ALT=[Maple Plot]"

PRIVATE "TYPE=PICT;ALT=[Maple Plot]"


solve the dynamic equation: start in the + chevron state. 
what voltage is necessary to switch to the - chevron state 
> Vsub:=-0.1;
sol:=dsolve({diff(phi(t),t)=subs({phi=phi(t),V=Vsub} union pars2,(1/mu)*torque),phi(0)=phichev},{phi(t)}, type=numeric, output=listprocedure);
odeplot(sol,[t,phi(t)],0..0.0001,labels=["time (secs)","phi (rads)"]); 

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

PRIVATE "TYPE=PICT;ALT=[Maple Math]"

PRIVATE "TYPE=PICT;ALT=[Maple Plot]"

> 
> 
> 
